MAD-X

Elias Métral

- **(Short) introduction to MAD-X code**
- **Download and execution of MAD-X code in local**
- **Simple examples: 1-2-3-4-5-6-7**
- **Example 8: Case of the SPS machine**

(SHORT) INTRODUCTION TO MAD-X CODE (1/2)

- **Want to design an accelerator? => Use MAD-X code**

- **MAD = Methodical Accelerator Design**

- **Homepage = http://cern.ch/mad (or http://mad.web.cern.ch/mad/)**

- **Documentation (MAD-X primer, user's guide, tutorials, etc.) = http://mad.web.cern.ch/mad/www/documentation.html**

- **Current responsible person: Laurent Deniau (Laurent.Deniau@cern.ch) => Support & questions to**
  - **MAD Team: mad@cern.ch**
  - **or MAD community: mad-usr@cern.ch**

  => Information web page: http://mad.web.cern.ch/mad/www/information.html

Likes feedback!

- **See recent nice Introduction by Guido Sterbini: https://indico.cern.ch/getFile.py/access?contribId=32&resId=0&materialId=slides&confId=218284**

(SHORT) INTRODUCTION TO MAD-X CODE (2/2)

- **MAD-X is a general purpose beam optics and lattice program distributed for free by CERN**

- **MAD-X is an optics program => Single particle dynamics**

- **It is used at CERN since more than 20 years for machine design and simulation (PS, SPS, LHC, linacs, etc.)**

- **MAD-X is written in C/C++/Fortran77/Fortran90 (source code is available under CERN copyright)**

DOWNLOAD AND EXECUTION OF MAD-X CODE IN LOCAL (1/2)

Procedure given in the "Releases" page:
http://mad.web.cern.ch/mad/www/releases.html

- 1) Go to general MAD-X web page: http://cern.ch/mad
- 2) Click on **Releases** (on the left)
- 3) Click on **development releases repository** (in the MAD-X development release) if it is for a MAC. Otherwise we should use the production release (above)
- 4) Right click on **madx-dev-macosx32** which is the executable to be used and save it in a folder (to be created and which we will call **MADXFolderLund2013**) where we will run MAD-X
- 5) Then open a Terminal and go to this folder
- 6) Type **mv madx-dev-macosx32 madx** to have a shorter name of the executable (called now madx)

DOWNLOAD AND EXECUTION OF MAD-X CODE IN LOCAL (2/2)

- 7) Make this file be an executable by typing: **chmod u+x madx**
- 8) Execute it to see if this works by typing **./madx** => One should see something like this

```
macbook-pro-de-elias-metral-7:MADXFolderLund2013 eliasmetral$ ./madx

  ++++++++++++++++++++++++++++++++++++++++++
  +     MAD-X 5.00.20  (32 bit, Darwin)    +
  +          Production Version            +
  + Code Modification Date: 2013.02.20     +
  + Execution Time Stamp: 23.02.13 13.50.08 +
  ++++++++++++++++++++++++++++++++++++++++++
X: ==>
```

MAD-X releases moves these days
=> Version 5.02.00 will soon be available

- 9) Type **quit;** to quit

```
quit;


  Number of warnings: 0

  ++++++++++++++++++++++++++++++++++++++++++
  + MAD-X 5.00.20 (32 bit) finished normally +
  ++++++++++++++++++++++++++++++++++++++++++
macbook-pro-de-elias-metral-7:MADXFolderLund2013 eliasmetral$
```

EXAMPLE 1 (1/4)

- 1st simple example => In a text editor, type the following lines and save the file in the folder MADXFolderLund2013 with the name Example1.mad

Example1.mad

```
/* Definition of the elements */
qfType:QUADRUPOLE, L=0.01, K1:=kf;
qdType:QUADRUPOLE, L=0.01, K1:=kd;
/* Definition of the the sequence */
fodo:SEQUENCE, REFER=entry, L=20;
qf1: qfType, at=0.0;
qd1: qdType, at=5.0;
qf2: qfType, at=10.0;
qd2: qdType, at=15.0;
ENDSEQUENCE;
/* Definition of the strengths */
kf=+20.0;
kd=-20.0;
/* Definition of the beam */
beam, particle=proton, energy=7000;
/* Activation of the sequence */
use, sequence=fodo;
/* Computations and plots */
twiss;
plot, HAXIS=s, VAXIS=betx, bety;
quit;
```

For comments: /*…*/ or start with // or !

Regular assignment

Deferred assignment: if kf changes, K1 is updated too

Length of the sequence

Can be: entry, centre or exit

MAD language is case insensitive

All statements are terminated with ;

EXAMPLE 1 (2/4)

- Type (in the folder MADXFolderLund2013) ./madx Example1.mad
- The 2 results should be
  - 1) Execution of the program

```
macbook-pro-de-elias-metral-7:MADXFolderLund2013 eliasmetral$ ./madx Example1.mad

  ++++++++++++++++++++++++++++++++++++++++++++
  +    MAD-X 5.00.20  (32 bit, Darwin)       +
  +          Production Version              +
  + Code Modification Date: 2013.02.20       +
  + Execution Time Stamp: 23.02.13 14.13.04 +
  ++++++++++++++++++++++++++++++++++++++++++++
/* Definition of the elements */

qfType:QUADRUPOLE, L=0.01, K1:=kf;

qdType:QUADRUPOLE, L=0.01, K1:=kd;

/* Definition of the the sequence */

fodo:SEQUENCE, REFER=entry, L=20;

qf1: qfType, at=0.0;

qd1: qdType, at=5.0;

qf2: qfType, at=10.0;

qd2: qdType, at=15.0;

ENDSEQUENCE;

/* Definition of the strengths */

kf=+20.0;

kd=-20.0;

/* Definition of the beam */

beam, particle=proton, energy=7000;

/* Activation of the sequence */

use, sequence=fodo;

/* Computations and plots */

twiss;
```

$$Q^{\text{Theory}} = 0.333$$

```
enter Twiss module
++++++ info: Zero value of SIGT replaced by 1.
++++++ info: Zero value of SIGE replaced by 1/1000.

iteration:   1 error:   0.000000E+00 deltap:   0.000000E+00
orbit:   0.000000E+00  0.000000E+00  0.000000E+00  0.000000E+00  0.000000E+00  0.000000E+00

++++++ table: summ

            length            orbit5              alfa           gammatr
                20                -0  -3.30872245e-24  -5.497558139e+11

                q1               dq1           betxmax             dxmax
      0.3330882356     -0.3672258596       17.31282068                 0

             dxrms            xcomax            xcorms                q2
                 0                 0                 0      0.3330882356

               dq2           betymax             dymax             dyrms
     -0.3672258596       17.31282068                -0                 0

            ycomax            ycorms            deltap           synch_1
                 0                 0                 0                 0

           synch_2           synch_3           synch_4           synch_5
                 0                 0                 0                 0
plot, HAXIS=s, VAXIS=betx, bety;

Plot - default table plotted: twiss
++++++ info: Zero value of SIGT replaced by 1.
++++++ info: Zero value of SIGE replaced by 1/1000.

 GXPLOT-X11  1.50 initialized

 plot number =            1
quit;

  Number of warnings: 0

  ++++++++++++++++++++++++++++++++++++++++++
  + MAD-X 5.00.20 (32 bit) finished normally +
  ++++++++++++++++++++++++++++++++++++++++++
macbook-pro-de-elias-metral-7:MADXFolderLund2013 eliasmetral$
```

$$Q'^{\text{Theory}} = -0.367$$

EXAMPLE 1 (3/4)

- 2) Creation of a new file called **madx.ps** => Open it with Ghostview for instance by typing **gv madx.ps &**



EXAMPLE 1 (4/4)

- **Comparison with expected theoretical values**
  - Length of a quadrupole (F and D) $l = 0.01 \text{ m}$
  - Strength of a quadrupole (F and D) $k = 20.0 \text{ m}^{-2}$
  - Total length of the FODO cell (2 L) $2\,L = 10 \text{ m}$
  - Number of cells $N_{cell} = 2$

$$f = f_F = 5 \text{ m}$$

$$\mu = 2 \arcsin\left(\frac{L}{2\,f}\right) = 1.05 \text{ rad} = 60 \text{ deg}$$

$$f_{F,D} = \pm \frac{1}{k\,l}$$

$$Q = \frac{N_{cell}\,\mu}{2\,\pi} = 0.33$$

$$\beta_{QF} = 2\,L\,\frac{1 + \sin\left(\frac{\mu}{2}\right)}{\sin(\mu)} = 17.3 \text{ m}$$

$$\beta_{QD} = 2\,L\,\frac{1 - \sin\left(\frac{\mu}{2}\right)}{\sin(\mu)} = 5.8 \text{ m}$$

$$\xi = -\frac{2}{\mu}\tan\left(\frac{\mu}{2}\right) = -1.1 = Q' / Q$$

EXAMPLE 2

- **Same as before but with the FODO cell length multiplied by 4**

EXAMPLE 3 (1/3)

- **Same as Example 1 but asking for a certain value of the tunes => Matching condition**

Can be done on global parameters (tune, chromaticity, etc.) or any local parameter

Example3.mad

```
/* Definition of the elements */
qfType:QUADRUPOLE, L=0.01, K1:=kf;
qdType:QUADRUPOLE, L=0.01, K1:=kd;
/* Definition of the the sequence */
fodo:SEQUENCE, REFER=entry, L=20;
qf1: qfType, at=0.0;
qd1: qdType, at=5.0;
qf2: qfType, at=10.0;
qd2: qdType, at=15.0;
ENDSEQUENCE;
/* Definition of the strengths */
kf=+20.0;
kd=-20.0;
/* Definition of the beam */
beam, particle=proton, energy=7000;
/* Activation of the sequence */
use, sequence=fodo;
/* Computations and plots */
twiss;
plot, HAXIS=s, VAXIS=betx, bety;
/*Matching*/
MATCH, sequence=fodo;
GLOBAL, Q1=.25;
GLOBAL, Q2=.25;
VARY, NAME=kf, STEP=0.00001;
VARY, NAME=kd, STEP=0.00001;
LMDIF, CALLS=50, TOLERANCE=1e-8;
ENDMATCH;
quit;
```

Method adopted

EXAMPLE 3 (2/3)

```
MATCH SUMMARY

Node_Name                 Constraint   Type  Target Value       Final Value        Penalty
--------------------------------------------------------------------------------------------------
Global constraint:        q1           4     2.50000000E-01     2.50002171E-01     4.71246224E-10
Global constraint:        q2           4     2.50000000E-01     2.50002640E-01     6.97000102E-10


Final Penalty Function =  1.16824633e-09





Variable                 Final Value  Initial Value Lower Limit  Upper Limit
--------------------------------------------------------------------------------
kf                       1.53177e+01  2.00000e+01 -1.00000e+20  1.00000e+20
kd                      -1.53177e+01 -2.00000e+01 -1.00000e+20  1.00000e+20

END MATCH SUMMARY
```

EXAMPLE 3 (3/3)

- **Comparison with expected theoretical values**

$$Q = \frac{N_{cell}\ \mu}{2\ \pi} = 0.25$$

=>

$$\mu = \frac{2\ \pi\ Q}{N_{cell}} = 0.785$$

=>

$$f = \frac{L}{2\ \sin\left(\frac{\mu}{2}\right)} = 6.53\ \mathrm{m}$$

=>

$$k = k_F = \frac{1}{f\ l} = 15.3\ \mathrm{m}$$

$$k_D = -\ k_F$$

EXAMPLE 4 (1/2)

- **Same as Example 1 but with** $k = k_F = \dfrac{1}{f\ l} = 15.3\ \mathrm{m}$ $k_D = -\,k_F$

Example4.mad

```
/* Definition of the elements */
qfType:QUADRUPOLE, L=0.01, K1:=kf;
qdType:QUADRUPOLE, L=0.01, K1:=kd;
/* Definition of the the sequence */
fodo:SEQUENCE, REFER=entry, L=20;
qf1: qfType, at=0.0;
qd1: qdType, at=5.0;
qf2: qfType, at=10.0;
qd2: qdType, at=15.0;
ENDSEQUENCE;
/* Definition of the strengths */
kf=+15.3;
kd=-15.3;
/* Definition of the beam */
beam, particle=proton, energy=7000;
/* Activation of the sequence */
use, sequence=fodo;
/* Computations and plots */
twiss;
plot, HAXIS=s, VAXIS=betx, bety;
quit;
```

EXAMPLE 4 (2/2)

length	orbit5	alfa	gammatr
20	-0	-3.30872245e-24	-5.497558139e+11
q1	dq1	betxmax	dxmax
0.2496978561	-0.2633426252	19.55650218	0
dxrms	xcomax	xcorms	q2
0	0	0	0.2496978561
dq2	betymax	dymax	dyrms
-0.2633426252	19.55650218	-0	0
ycomax	ycorms	deltap	synch_1
0	0	0	0
synch_2	synch_3	synch_4	synch_5
0	0	0	0

EXAMPLE 5 (1/3)

- Same as Example 1 but replacing the drifts by dipoles

Example5.mad

```
/* Definition of the elements */
qfType:QUADRUPOLE, L=0.01, K1:=kf;
qdType:QUADRUPOLE, L=0.01, K1:=kd;
MBType:RBEND, L=4.985, ANGLE:=ANGLE1;
/* Definition of the the sequence */
fodo:SEQUENCE, REFER=entry, L=20;
qf1: qfType, at=0.0;
MB1: MBType, at=0.01;
qd1: qdType, at=5.0;
MB2: MBType, at=5.01;
qf2: qfType, at=10.0;
MB3: MBType, at=10.01;
qd2: qdType, at=15.0;
MB4: MBType, at=15.01;
ENDSEQUENCE;
/* Definition of the strengths */
kf=+20.0;
kd=-20.0;
ANGLE1:=0.1;
/* Definition of the beam */
beam, particle=proton, energy=7000;
/* Activation of the sequence */
use, sequence=fodo;
/* Computations and plots */
twiss;
plot, HAXIS=s, VAXIS=betx, bety, dx, dy;
quit;
```

RBEND = Rectangular BENDing magnet. There is also SBEND = Sector BENDing magnet

EXAMPLE 5 (2/3)



gv: madx.ps
File
State
Page
Landscape
1.000
BBox
madx.ps
2013/05/20 13:52:23
Variable Size
Open
Print All
Print Marked
Save All
Save Marked
Reload
Dipole
Q_F
Q_D
fodo
MAD-X 5.00.20 20/05/13 13.52.23
β_x
β_y
D_x
D_y
20.
18.
16.
14.
12.
10.
8.
6.
4.
2.
0.0
0.0
2.
4.
6.
8.
12.
14.
16.
18.
20.
s (m)
$D_{max}^{Theory} = D_{QF} = 2.5$ m
$D_{min}^{Theory} = D_{QD} = 1.5$ m

EXAMPLE 5 (3/3)

- **Comparison with expected theoretical values**

$$f = f_F = 5 \text{ m}$$

(See Example 1)

$$\rho_0 = \frac{l_d}{\vartheta_d} = \frac{4.985}{0.1} = 49.85$$

$$D_{QF} = \frac{4 f^2}{\rho_0} \left(1 + \frac{L}{4 f} \right) \approx 2.5 \text{ m}$$

$$D_{QD} = \frac{4 f^2}{\rho_0} \left(1 - \frac{L}{4 f} \right) \approx 1.5 \text{ m}$$

EXAMPLE 6 (1/2)

- **Same as Example 5 but introducing some thin sextupoles at the locations of the quadrupoles (2 families) to correct the chromaticities to 0**

  => Numerical values for the sextupole strengths deduced from theory (see page 69 of course on TBD):

$$\text{ksf} = \frac{1}{2} \times (-) \frac{4\pi}{D_{QF}\; l_s} \left(\frac{\beta_{QF}\, Q_x\, \xi_x + \beta_{QD}\, Q_y\, \xi_y}{\beta_{QF}^2 - \beta_{QD}^2} \right) \approx 7.41 \text{ m}^{-3}$$

$$\text{ksd} = \frac{1}{2} \times \frac{4\pi}{D_{QD}\; l_s} \left(\frac{\beta_{QD}\, Q_x\, \xi_x + \beta_{QF}\, Q_y\, \xi_y}{\beta_{QF}^2 - \beta_{QD}^2} \right) \approx -12.56 \text{ m}^{-3}$$

As there are 2 sextupoles / family

Exercise6.mad

```
/* Definition of the elements */
qfType:QUADRUPOLE, L=0.01, K1:=kf;
qdType:QUADRUPOLE, L=0.01, K1:=kd;
MBType:RBEND, L=4.975, ANGLE:=ANGLE1;
SfType:SEXTUPOLE, L=0.01, K2:=ksf;
SdType:SEXTUPOLE, L=0.01, K2:=ksd;
/* Definition of the the sequence */
fodo:SEQUENCE, REFER=entry, L=20;
qf1: qfType, at=0.0;
Sf1: SfType, at=0.01;
MB1: MBType, at=0.02;
qd1: qdType, at=5.0;
Sd1: SdType, at=5.01;
MB2: MBType, at=5.02;
qf2: qfType, at=10.0;
Sf2: SfType, at=10.01;
MB3: MBType, at=10.02;
qd2: qdType, at=15.0;
Sd2: SdType, at=15.01;
MB4: MBType, at=15.02;
ENDSEQUENCE;
/* Definition of the strengths */
kf=+20.0;
kd=-20.0;
ANGLE1=0.1;
ksf=7.41;
ksd=-12.56;
/* Definition of the beam */
beam, particle=proton, energy=7000;
/* Activation of the sequence */
use, sequence=fodo;
/* Computations and plots */
twiss;
plot, HAXIS=s, VAXIS=betx, bety, dx, dy;
quit;
```

EXAMPLE 6 (2/2)

```
++++++ table: summ

       length           orbit5             alfa          gammatr
           20               -0    0.03934850023      5.041222989

           q1              dq1          betxmax            dxmax
  0.332479065  -4.61347819e-05      17.29373077      2.507673771

        dxrms           xcomax           xcorms               q2
  2.121898591                0                0     0.3404086714

          dq2          betymax            dymax            dyrms
-0.002109151774      17.03224584                0                0

       ycomax           ycorms           deltap          synch_1
            0                0                0                0

      synch_2          synch_3          synch_4          synch_5
            0                0                0                0
```

The chromaticities are indeed very close to 0

EXAMPLE 7 (1/2)

- **Same as Example 6 but finding the required sextupole strengths by matching**

Exercise7.mad

```
/* Definition of the elements */
qfType:QUADRUPOLE, L=0.01, K1:=kf;
qdType:QUADRUPOLE, L=0.01, K1:=kd;
MBType:RBEND, L=4.975, ANGLE:=ANGLE1;
SfType:SEXTUPOLE, L=0.01, K2:=ksf;
SdType:SEXTUPOLE, L=0.01, K2:=ksd;
/* Definition of the the sequence */
fodo:SEQUENCE, REFER=entry, L=20;
qf1: qfType, at=0.0;
Sf1: SfType, at=0.01;
MB1: MBType, at=0.02;
qd1: qdType, at=5.0;
Sd1: SdType, at=5.01;
MB2: MBType, at=5.02;
qf2: qfType, at=10.0;
Sf2: SfType, at=10.01;
MB3: MBType, at=10.02;
qd2: qdType, at=15.0;
Sd2: SdType, at=15.01;
MB4: MBType, at=15.02;
ENDSEQUENCE;
/* Definition of the strengths */
kf=+20.0;
kd=-20.0;
ANGLE1=0.1;
ksf=0;
ksd=0;
/* Definition of the beam */
beam, particle=proton, energy=7000;
/* Activation of the sequence */
use, sequence=fodo;
/* Computations and plots */
twiss;
plot, HAXIS=s, VAXIS=betx, bety, dx,
dy;
/*Matching*/
MATCH, sequence=fodo;
GLOBAL, dQ1=0.0;
GLOBAL, dQ2=0.0;
VARY, NAME=ksf, STEP=0.00001;
VARY, NAME=ksd, STEP=0.00001;
LMDIF, CALLS=50, TOLERANCE=1e-8;
ENDMATCH;
quit;
```

EXAMPLE 7 (2/2)

```
MATCH SUMMARY

Node_Name                 Constraint   Type  Target Value       Final Value        Penalty
--------------------------------------------------------------------------------------------------
Global constraint:        dq1          4     0.00000000E+00    -8.40104445E-12     7.05775479E-23
Global constraint:        dq2          4     0.00000000E+00    -3.23113830E-12     1.04402547E-23


Final Penalty Function =   8.10178026e-23



Variable                 Final Value  Initial Value Lower Limit  Upper Limit
--------------------------------------------------------------------------------
ksf                      7.42249e+00  0.00000e+00 -1.00000e+20  1.00000e+20
ksd                     -1.26185e+01  0.00000e+00 -1.00000e+20  1.00000e+20

END MATCH SUMMARY
```

Close to theoretical values

EXAMPLE 8 (1/13)

- Case of the SPS machine (in 2008) without or with extraction bump

=> Info split into several files (due to large number of elements)

- sps2008.ele — Elements
- sps2008.seq — Sequence
- elements.str — Strengths
- ft_qs_ext_2008.str — Strengths
- CNGS_extraction.beamx — Type of beam
- twiss.cmdx — Command file
- general.madx — Type ./madx general.madx to execute the program (general.madx calls all the other files)

EXAMPLE 8 (2/13)

- **sps2008.ele**

```
ACL       : RFCAVITY     , L := 3.52913  ;
ACTA      : RFCAVITY     , L := 17.782   ;
ACTB      : RFCAVITY     , L := 17.782   ;
ACTC      : RFCAVITY     , L := 21.896   ;
ACTD      : RFCAVITY     , L := 21.896   ;
AEG       : INSTRUMENT   , L := .5593    ;
AEP       : INSTRUMENT   , L := 1        ;
AEPA      : INSTRUMENT   , L := .722     ;
AEPC      : INSTRUMENT   , L := .722     ;
AERB      : HMONITOR     , L := .685     ;
AESA      : INSTRUMENT   , L := .2305    ;
AETA      : INSTRUMENT   , L := .21      ;
AEW       : INSTRUMENT   , L := .5105    ;
AEWA      : INSTRUMENT   , L := .5105    ;
AEWB      : INSTRUMENT   , L := .5661    ;
APWL      : INSTRUMENT   , L := .776     ;
BBLR      : MONITOR      , L := .72      ;
BBLRM     : MONITOR      , L := .8       ;
BBSR      : INSTRUMENT   , L := .41      ;
BCT       : INSTRUMENT   , L := .694     ;
BDH       : HMONITOR     , L := 2.32     ;
BDV       : VMONITOR     , L := 1.46     ;
BFCT      : INSTRUMENT   , L := .482     ;
BMO       : MONITOR      , L := .45      ;
BPCE      : MONITOR      , L := .4       ;
BPCL      : MONITOR      , L := .72      ;
BPCN      : MONITOR      , L := .236     ;
BPCR      : MONITOR      , L := .598     ;
BPD       : MONITOR      , L := .4       ;
BPH       : HMONITOR     , L := .275     ;
BPHA      : MONITOR      , L := .5       ;
BPL       : MONITOR      , L := .45      ;
BPLT      : MONITOR      , L := .681     ;
BPMBH     : MONITOR      , L := .13      ;
BPMBV     : MONITOR      , L := .13      ;
BPNB      : HMONITOR     , L := .364     ;
BPNC      : HMONITOR     , L := .562     ;
BPSH      : HMONITOR     , L := 3.38     ;
BPSV      : VMONITOR     , L := 3.38     ;
BPV       : VMONITOR     , L := .275     ;
BPVA      : MONITOR      , L := .5       ;
BPVC      : VMONITOR     , L := .562     ;
BPVT      : VMONITOR     , L := .4       ;
BPW       : VMONITOR     , L := .86      ;
BRCH      : RCOLLIMATOR  , L := .6       ;
```

...

```
ENDOFSPS  : MARKER       ;
LOD       : OCTUPOLE     , L := .677     ;
LODOUT    : OCTUPOLE     , L := .677     ;
LOE       : OCTUPOLE     , L := .74      ;
LOEN      : OCTUPOLE     , L := .656     ;
LOF       : OCTUPOLE     , L := .705     ;
LOFOUT    : OCTUPOLE     , L := .705     ;
LQS       : QUADRUPOLE   , L := .542     ;
LSD       : SEXTUPOLE    , L := .42      ;
LSE       : SEXTUPOLE    , L := .74      ;
LSEN      : SEXTUPOLE    , L := .72      ;
LSF       : SEXTUPOLE    , L := .423     ;
MBA       : RBEND        , L := 6.26     ;
MBB       : RBEND        , L := 6.26     ;
MDH       : HKICKER      , L := .25      ;
MDHA      : HKICKER      , L := .306     ;
MDHB      : HKICKER      , L := .402     ;
MDHD      : HKICKER      , L := .36      ;
MDHW      : HKICKER      , L := .526     ;
MDPH      : HKICKER      , L := .25      ;
MDSH      : HKICKER      , L := .7       ;
MDV       : VKICKER      , L := .25      ;
MDVA      : VKICKER      , L := .424     ;
MDVB      : VKICKER      , L := .41      ;
MDVW      : VKICKER      , L := .526     ;
MKDHA     : HKICKER      , L := 1.6      ;
MKDHB     : HKICKER      , L := 1.6      ;
MKDVA     : VKICKER      , L := 2.892    ;
MKDVB     : VKICKER      , L := 2.892    ;
MKEL      : HKICKER      , L := 2.014    ;
MKES      : HKICKER      , L := 2.014    ;
MKP       : HKICKER      , L := 3.423    ;
MKPA      : HKICKER      , L := 3.423    ;
MKPC      : HKICKER      , L := 1.78     ;
MKQH      : HKICKER      , L := .96      ;
MKQV      : VKICKER      , L := 1.416    ;
MPLH      : HKICKER      , L := 2.06     ;
MPLV      : VKICKER      , L := .555     ;
MPNH      : HKICKER      , L := 2.04     ;
MPSH      : HKICKER      , L := .714     ;
MPSV      : VKICKER      , L := .275     ;
MSE       : HKICKER      , L := 2.28     ;
MST       : HKICKER      , L := 2.40     ;
QE        : QUADRUPOLE   , L := .698     ;
QM        : QUADRUPOLE   , L := 3.085    ;
QMA       : QUADRUPOLE   , L := 3.791    ;
QMS       : QUADRUPOLE   , L := .705     ;
QSE       : QUADRUPOLE   , L := .42      ;
QSPLIT    : MARKER       ;
STARTSPS  : MARKER       ;
TBSJB     : RCOLLIMATOR  , L := 2.1      ;
```

...

EXAMPLE 8 (3/13)

- **sps2008.seq**

```
option, echo;

SPS: SEQUENCE, L = 6911.5038;

BEGI.10010          :  STARTSPS    , AT := 0          ;
QF.10010            :  QF2.F       , AT := 1.5425     ;
MBA.10030           :  MBA.F       , AT := 6.575      ;
MBA.10050           :  MBA.F       , AT := 13.235     ;
MBB.10070           :  MBB.F       , AT := 19.885     ;
MBB.10090           :  MBB.F       , AT := 26.525     ;
VVSA.10101          :  VVSA        , AT := 29.9385    ;
LSD.10105           :  LSDA.F      , AT := 31.0007    ;
MDV.10107           :  MDV10107.F  , AT := 31.4877    ;
BPV.10108           :  BPV         , AT := 31.7652    ;
QD.10110            :  QD.F        , AT := 33.5402    ;
MBB.10130           :  MBB.F       , AT := 38.5627    ;
MBB.10150           :  MBB.F       , AT := 45.2027    ;
MBA.10170           :  MBA.F       , AT := 51.8527    ;
MBA.10190           :  MBA.F       , AT := 58.5127    ;
LSF.10205           :  LSFB.F      , AT := 62.9984    ;
MDH.10207           :  MDH10207.F  , AT := 63.4754    ;
BPH.10208           :  BPH         , AT := 63.7629    ;
QF.10210            :  QF2.F       , AT := 65.5379    ;
```

...

```
QF.63410            :  QF2.F       , AT := 6849.0509  ;
MBA.63430           :  MBA.F       , AT := 6854.0834  ;
MBA.63450           :  MBA.F       , AT := 6860.7434  ;
MBB.63470           :  MBB.F       , AT := 6867.3934  ;
MBB.63490           :  MBB.F       , AT := 6874.0334  ;
LOD.63502           :  LOD.F       , AT := 6877.8004  ;
LSD.63505           :  LSDB.F      , AT := 6878.5091  ;
MDV.63507           :  MDV63507.F  , AT := 6878.9961  ;
BPV.63508           :  BPV         , AT := 6879.2736  ;
QD.63510            :  QD.F        , AT := 6881.0486  ;
MBB.63530           :  MBB.F       , AT := 6886.0711  ;
MBB.63550           :  MBB.F       , AT := 6892.7111  ;
MBA.63570           :  MBA.F       , AT := 6899.3611  ;
MBA.63590           :  MBA.F       , AT := 6906.0211  ;
LOE.63602           :  LOE63602.F  , AT := 6909.8401  ;
LSF.63605           :  LSFA.F      , AT := 6910.5088  ;
MDH.63607           :  MDH63607.F  , AT := 6910.9838  ;
BPH.63608           :  BPH         , AT := 6911.2713  ;
END.10010           :  ENDOFSPS    , AT := 6911.5038  ;

ENDSEQUENCE;

option, -echo;

return;
```

EXAMPLE 8 (4/13)

- **elements.str**

```
option, -echo;

LSDA.F       :   LSD          ,  K2     := ksda;
LSDB.F       :   LSD          ,  K2     := ksdb;
LSFA.F       :   LSF          ,  K2     := ksfa;
LSFB.F       :   LSF          ,  K2     := ksfb;
LSFC.F       :   LSF          ,  K2     := KSFC;
MBA.F        :   MBA          ,  ANGLE := 8.445141542E-03;
MBB.F        :   MBB          ,  ANGLE := 8.445141542E-03;
QD.F         :   QM           ,  K1     := KQD;
QDA.F        :   QMA          ,  K1     := KQDA;
QF1.F        :   QM           ,  K1     := KQF1;
QF1A.F       :   QMA          ,  K1     := KQF1A;
QF2.F        :   QM           ,  K1     := KQF2;
QF2A.F       :   QMA          ,  K1     := KQF2A;
ACL31695.F   :   ACL          ,  VOLT   := VACL31695;
ACL31733.F   :   ACL          ,  VOLT   := VACL31733;
ACTA31637.F  :   ACTA         ,  VOLT   := VACTA31637, FREQ=200.266; !, HARMON=4620;
ACTB31739.F  :   ACTB         ,  VOLT   := VACTB31739, FREQ=200.266; !, HARMON=4620;
ACTC31836.F  :   ACTC         ,  VOLT   := VACTC31836, FREQ=200.266; !, HARMON=4620;
ACTD31934.F  :   ACTD         ,  VOLT   := VACTD31934, FREQ=200.266; !, HARMON=4620;
BRCH51902.F  :   BRCH         ,  XSIZE := 0.00;
BRCV51899.F  :   BRCV         ,  XSIZE := 0.00;
LOD.F        :   LOD          ,  K3     := K3LOD;
```

...

```
ZKHA21991.F  :   ZKHA         ,  KICK   := KZKHA21991;
ZKV21993.F   :   ZKV          ,  KICK   := KZKV21993;
ZS21633.F    :   ZS           ,  KICK   := KZS21633;
ZS21638.F    :   ZS           ,  KICK   := KZS21638;
ZS21655.F    :   ZS           ,  KICK   := KZS21655;
ZS21671.F    :   ZS           ,  KICK   := KZS21671;
ZS21676.F    :   ZS           ,  KICK   := KZS21676;

option, echo;

return;
```

EXAMPLE 8 (5/13)

- **ft_qs_ext_2008.str**

```
option, -echo;

ksda:=-0.1010244459;
ksdb:=-0.1439436811;
ksfa:=0.02344776222;
ksfb:=0.000211064706;
KSFC:=KSFA;

KQD:=-0.01462465119;
KQDA:=KQD*9.0/11.0;
KQF1:=0.01454253978;
KQF1A:=KQF1*9.0/11.0;
KQF2:=0.01478155121;
KQF2A:=KQF2*9.0/11.0;

! User definable strengths
VACL31695        := 0.00;
VACL31733        := 0.00;
VACTA31637       := 0.00;
VACTB31739       := 0.00;
VACTC31836       := 0.00;
VACTD31934       := 0.00;
K3LOD            := 0.00;
```

...

```
KMPLH41658       := 0.0004795319*On_ExtBump;
KMPLH41994       := 0.0003673919*On_ExtBump;
KMPLH61655       := 0.00;
KMPLH61996       := 0.00;
KMPLV41501       := 0.0000253846*On_ExtBump;
KMPLV42101       := 0.0000006151*On_ExtBump;
KMPNH21732       := 0.00;
KMPSH21202       := 0.00;
KMPSH41402       := 0.0000439373*On_ExtBump;
KMPSH42198       := 0.0000686338*On_ExtBump;
KMPSH61402       := 0.00;
KMPSH62199       := 0.00;
KMPSV21303       := 0.00;
KMPSV21503       := 0.00;
KMPSV22103       := 0.00;
KMPSV22303       := 0.00;
KMPSV41303       := 0.000002044*On_ExtBump;
KMPSV42303       := -0.0000253862*On_ExtBump;
KMPSV61303       := 0.00;
```

...

```
K1QSE51897       := 0.00;
KZKHA21991       := 0.00;
KZKV21993        := 0.00;
KZS21633         := 0.00;
KZS21638         := 0.00;
KZS21655         := 0.00;
KZS21671         := 0.00;
KZS21676         := 0.00;

! END-OF-STRENGTHS

option, echo;
return;
```

EXAMPLE 8 (6/13)

- **CNGS_extraction.beamx**

```
Beam, particle = proton, pc = 400.0,exn=12.0e-6*4.0, eyn=7.0E-6*4.0,
sige=0.4e-3, NPART=2E10, BUNCHED;

Z:=1;
A:=1;
TBUNCH:=0.5e-9;
DPP:=BEAM->SIGE*(BEAM->ENERGY/BEAM->PC)^2;

value, dpp;
return;
```

EXAMPLE 8 (7/13)

- **twiss.cmdx**

```
select, flag=twiss, clear=true;
select, flag=twiss, range=#S/#E, column=name,s,x,y;

TWISS, DELTAP=0.0;
write, table=twiss, file=twiss.prt;

plot, haxis=s, vaxis=betx,bety, hmin= StartPlot, hmax= EndPlot, vmin=0.0, vmax=150.0, colour=100;
plot, haxis=s, vaxis=dx,dy, hmin= StartPlot, hmax= EndPlot, vmin=-1.0, vmax=6.0, colour=100;
plot, haxis=s, vaxis=mux,muy, hmin= StartPlot, hmax= EndPlot, vmin=0.0, vmax=30.0, colour=100;
plot, haxis=s, vaxis=x,y, hmin= StartPlot, hmax= EndPlot, vmin=-0.010, vmax=0.040, colour=100;

return;
```

EXAMPLE 8 (8/13)

- **general.madx**

```
 option, -echo;
 option, RBARC=FALSE;  ! the length of a rectangular magnet
                       ! is the distance between the polefaces
                       ! and not the arc length
 call, file = 'CNGS_extraction.beamx';

 call, file = 'sps2008.ele';

 call, file = 'elements.str';

 call, file = 'ft_qs_ext_2008.str';

 call, file = 'sps2008.seq';

option, echo;

! To plot only the 4th sextant
! StartPlot:=3456.0;
! EndPlot:=4608.0;
! StartPlot:=1020.0;
! EndPlot:=6060.0;
StartPlot:=0.0;
EndPlot:=6911.5038;

! Flag for the extraction bump
On_ExtBump:=0;

USE, period=SPS, range=#S/#E;

call, file = 'twiss.cmdx';
stop;
```

0 => Extraction bump OFF
1 => Extraction bump ON

EXAMPLE 8 (9/13)

- Typing ./max general.madx, yields
  - A file with all the data => twiss.prt
  - A file with all the plots => madx.ps

EXAMPLE 8 (10/13)

- Case with On_ExtBump:=0 => Extraction bump OFF







EXAMPLE 8 (11/13)





EXAMPLE 8 (12/13)

```
@ NAME             %05s "TWISS"
@ TYPE             %05s "TWISS"
@ SEQUENCE         %03s "SPS"
@ PARTICLE         %06s "PROTON"
@ MASS             %le        0.938272013
@ CHARGE           %le                  1
@ ENERGY           %le        400.0011004
@ PC               %le                400
@ GAMMA            %le        426.3167769
@ KBUNCH           %le                  1
@ BCURRENT         %le    0.0001389912007
@ SIGE             %le             0.0004
@ SIGT             %le                  0
@ NPART            %le              2e+10
@ EX               %le    2.814816039e-08
@ EY               %le    1.641976023e-08
@ ET               %le                  1
@ LENGTH           %le          6911.5038
@ ALFA             %le     0.001848026402
@ ORBIT5           %le                 -0
@ GAMMATR          %le        23.26193909
@ Q1               %le        26.66659956
@ Q2               %le        26.58000001
@ DQ1              %le       -36.24716692
@ DQ2              %le       0.6445710371
@ DXMAX            %le        5.271538934
@ DYMAX            %le                  0
@ XCOMAX           %le                  0
@ YCOMAX           %le                  0
@ BETXMAX          %le        103.9512232
@ BETYMAX          %le        106.5962215
@ XCORMS           %le                  0
@ YCORMS           %le                  0
@ DXRMS            %le        2.176283147
@ DYRMS            %le                  0
@ DELTAP           %le                  0
@ SYNCH_1          %le                  0
@ SYNCH_2          %le                  0
@ SYNCH_3          %le                  0
@ SYNCH_4          %le                  0
@ SYNCH_5          %le                  0
@ TITLE            %08s "no-title"
@ ORIGIN           %23s "MAD-X 5.00.20 Darwin 32"
@ DATE             %08s "20/05/13"
@ TIME             %08s "15.20.32"
* NAME                           S                  X                  Y
$ %s                           %le                %le                %le
 "SPS$START"                     0                  0                  0
 "BEGI.10010"                    0                  0                  0
 "QF.10010"                  3.085                  0                  0
 "DRIFT_0"                   3.445                  0                  0
 "MBA.10030"                 9.705                  0                  0
```

File twiss.prt

...

EXAMPLE 8 (13/13)

- **Case with On_ExtBump:=1 => Extraction bump ON**





Extraction bump not fully closed here